MAGMA CODES FOR "SOME SINGULAR CURVES AND SURFACES ARISING FROM INVARIANTS OF COMPLEX REFLECTION GROUPS"

by

CÉDRIC BONNAFÉ

Abstract. — This file contains the `Magma` codes justifying the results of [**Bon2**].

In [**Bon2**], the author constructs singular curves and surfaces arising from invariants of complex reflection groups by using extensive computations (see also [**Bon1**]). For the smoothness of the exposition of the paper [**Bon2**], the codes (written with the software `Magma` [**Magma**]) are not included there, but are given here, so that the reader can check every detail.

We do not intend to publish the present text: it will remain on `arXiv`(1), on `HAL`(2) and on the webpage of the author(3). We also take opportunity of this paper to give explicit equations for our varieties (some are very big, and could not be included in a published paper) as well as some pictures (generally made with `SURFER` [**Sur**]).

As mathematical details are mostly given in [**Bon2**] and the content of the present text is rather boring, the exposition here will be very rough (but hopefully rigorous). In the first sections, we give the part of the code which is common to all complex reflection groups and then every group will be treated separately in the upcoming sections.

Caution. For defining the exceptional complex reflection groups, we do not use the command `ShephardTodd` of MAGMA: the reason is that the model chosen by MAGMA is generally not Galois invariant and not only leads to ugly invariant polynomials, but leads to very long computations that sometimes cannot finish. We prefer using the models implemented in the file `primitive-complex-reflection-groups.m` which is contained in Appendix I. This file is based on a file created by Thiel in his `CHAMP` package [**Thi**]: these models are copy of models defined by Jean Michel in the `Chevie` package of `GAP3` (see [**Mic**]). We have just changed the groups $G_{23} = \mathrm{W}(H_3)$, $G_{28} = \mathrm{W}(F_4)$ and $G_{30} = \mathrm{W}(H_4)$ (see [**Bon2**, Remark 1.3] for a justification of this choices. With Thiel's package, the complex reflection group G_k (for $4 \leqslant k \leqslant 37$) is obtained through the command `ExceptionalComplexReflectionGroup(k)`. With our file, the group G_k is obtained through the following command:

```
> load 'primitive-complex-reflection-groups.m';
> W:=PrimitiveComplexReflectionGroup(k);
```

The author is partly supported by the ANR (Project No ANR-16-CE40-0010-01 GeRepMod).

(1)`https://arxiv.org/`

(2)`https://hal.archives-ouvertes.fr/`

(3)`http://imag.umontpellier.fr/~bonnafe/`

Commentary. The author is not an expert in computer programming and asks for some indulgence. Even though the codes might look inefficient/dirty for an expert, we hope that they are at least correct.

Auxiliary functions. We also needed to define some auxiliary functions for computing orbits under W in the projective space $\mathbf{P}(V)$, and determining the nature of the singularities. This is contained in a file `auxiliary-function.m` that can be downloaded by the reader in Appendix II.

Remark. All informations given in this remark are concerned with computations made using the computing facilities of MSRI and MAGMA version `Magma V2.23-7`. There are only four places where computations can be somewhat long:

- *Computing* $\phi(\mathscr{X}_{\text{sfib}})$. It takes less than 2 seconds for $k \in \{23, 24, 5, 26, 27, 28, 29\}$ but:
  - It takes about 20 seconds for $k = 30$ ($G_{30} = \mathrm{W}(H_4)$).
  - It takes about 14 minutes for $k = 31$.
  - It takes about 5 minutes for $k = 32$.
  - It takes about 47 minutes for $k = 33$.
  - It takes about 33 minutes for $k = 35$ ($G_{35} = \mathrm{W}(E_6)$).
  - For $k \in \{34, 36, 37\}$, the computation gives no result after a few hours and we have given up studying these cases.
- *Checking irreducibility over* $\mathbb{C}$. It takes less than 3 seconds for $k \in \{23, 25, 26, 28, 29, 30\}$, less than 20 seconds for $k \in \{31, 32\}$ but:
  - It takes about 17 minutes for $k = 24$. It is much more difficult to check it for a curve than for hypersurfaces of higher dimension, because of the trick involving Lemma 1.1.
  - It takes about 40 seconds for $k = 27$.
  - It takes about 80 seconds for $k = 35$.
- *Computing the reduced singular locus of all the* $\mathscr{Z}(F_{u_i})$. It is almost immediate in the case of curves (i.e. $23 \leqslant k \leqslant 27$) and for $k \in \{28, 29\}$.
  - It takes about 20 seconds for $k = 30$.
  - It takes about 30 seconds for $k = 31$.
  - It takes about 7 minutes for $k = 32$.
  - It takes about 20 minutes for $k = 35$.
- *Computing the type of the singularities.*

Let us give the total time of all computations according to the value of k.

- $k = 23 \longrightarrow$ about 1 second.
- $k = 24 \longrightarrow$ about 18 minutes (most of the time is for checking absolute irreducibility).
- $k = 25 \longrightarrow$ about 3 seconds
- $k = 27 \longrightarrow$ about 50 seconds
- $k = 28 \longrightarrow$ about 1 second.
- $k = 29 \longrightarrow$ about 8 second.
- $k = 30 \longrightarrow$ about 4 minutes ad 20 seconds (most of the time is for defining $\mathscr{Z}(F_{u_i})$ as a `Surface` in MAGMA).
- $k = 31 \longrightarrow$ about 19 minutes
- $k = 32 \longrightarrow$ about 47 minutes. Here, it must be noticed that we have slightly modified the program so that it does not compute the Milnor and Tjurina numbers of the last surface: the program does not answer after few hours (the Milnor and Tjurina numbers have been computed with `Singular`).
- $k = 33 \longrightarrow$ about 44 minutes (this is only for computing the set $U_{\text{sing}}^{\text{irr}}$: as it is empty, there is nothing more to compute...).
- $k = 35 \longrightarrow$ about 59 minutes

1. General code

As we will be interested in the singular locus of projective hypersurfaces, recall the following very easy fact:

Lemma 1.1. — *Let $f \in \mathbb{C}[V]$ be homogeneous. If* $\dim \mathscr{Z}_{\mathrm{sing}}(f) \leqslant n-4$, *then* $\mathscr{Z}(f)$ *is irreducible.*

Proof. — Indeed, if $\mathscr{Z}(f)$ has (at least) two distinct irreducible components $\mathscr{Z}_1$ and $\mathscr{Z}_2$, then $\dim(\mathscr{Z}_1 \cap \mathscr{Z}_2) = n-3$ (because it is the intersection of two hypersurfaces in the projective space $\mathbf{P}^{n-1}(\mathbb{C})$), and all the points in $\mathscr{Z}_1 \cap \mathscr{Z}_2$ are singular. □

We use the notation of [**Bon2**]. Almost all our computations are made with `Magma` [**Magma**][(4)] The file `singular-varieties.m` below contains the `Magma` code common to all cases:

[(4)]Some Milnor and Tjurina numbers were computed with SINGULAR [**DGPS**].

```
load 'auxiliary-functions.m';
load 'primitive-complex-reflection-groups.m';
W:=PrimitiveComplexReflectionGroup(k);
if CoefficientRing(W) eq RationalField() then
  K:=RationalField();
else
  L<zeta>:=CoefficientRing(W);
  K<a>:=sub<L | &cat [Minors(w,1) : w in W]>;
end if;
W:=ChangeRing(W,K);
n:=Rank(W);
m:=n-1;
R:=InvariantRing(W);
if n eq 3 then
  P<x,y,z>:=PolynomialRing(R);
elif n eq 4 then
  P<x,y,z,t>:=PolynomialRing(R);
else
  P<[x]>:=PolynomialRing(R);
end if;
Pm:=Proj(P);
if k in [23,28,35,36] then
  dr:=6;
  f1:=InvariantsOfDegree(W,2)[1];
  f:=f1^3;
elif k eq 24 then
  dr:=14;
  f1:=InvariantsOfDegree(W,4)[1];
  f2:=InvariantsOfDegree(W,6)[1];
  f:=f1^2*f2;
elif k in [25,26,27,34] then
  dr:=12;
  f1:=InvariantsOfDegree(W,6)[1];
  f:=f1^2;
elif k eq 29 then
  dr:=8;
  f1:=InvariantsOfDegree(W,4)[1];
  f:=f1^2;
elif k eq 30 then
  dr:=12;
  f1:=InvariantsOfDegree(W,2)[1];
  f:=f1^6;
elif k eq 31 then
  dr:=20;
  f1:=InvariantsOfDegree(W,8)[1];
  f2:=InvariantsOfDegree(W,12)[1];
  f:=f1*f2;
elif k eq 32 then
  dr:=24;
  f1:=InvariantsOfDegree(W,12)[1];
  f:=f1^2;
elif k eq 33 then
  dr:=10;
  f1:=InvariantsOfDegree(W,4)[1];
  f2:=InvariantsOfDegree(W,6)[1];
  f:=f1*f2;
```

```
elif k eq 37 then
  dr:=8;
  f1:=InvariantsOfDegree(W,4)[1];
  f:=f1^2;
end if;
fr:=InvariantsOfDegree(W,dr)[1];
if Gcd(fr,f) ne 1 then
  print "The two invariants of degree d_r are not linearly independent,
  so the computation must stop";
else
  print "fr is a fundamental invariant";
  AmxA1<[u]>:=AffineSpace(K,n);
  A1:=AffineSpace(K,1);
  phi:=map<AmxA1->A1 | [u[n]]>;
  affinization:=function(p)
    return Evaluate(p,[u[i] : i in [1..n-1]] cat [1]);
  end function;
  faff:=affinization(f);
  fraff:=affinization(fr);
  Fuaff:=fraff + u[n] * faff;
  X:=Scheme(AmxA1,Fuaff);
  Xsfib:=Scheme(X,[Derivative(Fuaff,i) : i in [1..n-1]]);
  print "Computing phi(Xsfib)...";
  time Psing:=MinimalBasis(phi(Xsfib));
 if # Psing eq 0 then print
   "There is no singular variety of the form Z(F_u)";
 else
  if Set([Degree(i[1]) : i in Factorization(Psing[1])]) ne {1} then
    print "One must extend the field of definition";
  else
    print "All elements of U_sing belong to K";
    Using:=[-Evaluate(i[1],[0]) : i in Factorization(Psing[1])];
    Am<[x]>:=AffineSpace(K,n-1);
    Usingirr:=[ui : ui in Using |
        IsIrreducible(Scheme(Am,Evaluate(Fuaff,x cat [ui])))];
    Zaff:=[Scheme(Am,Evaluate(Fuaff,x cat [ui])) : ui in Usingirr];
    print "Checking absolute irreducibility...";
    if n eq 3 then
      time set:=Set([IsAbsolutelyIrreducible(Curve(var)) : var in Zaff]);
      if set ne {true} then
        print "There is a non-absolutely irreducible curve";
      end if;
      Z:=[Curve(Pm,fr+ui*f) : ui in Usingirr];
    else
      time number:=# [var : var in Zaff |
        Dimension(SingularSubscheme(var)) gt n-4];
      if number gt 0 then
        print "There is possibly a non-absolutely irreducible variety";
      end if;
      Fu:=[fr+ui*f : ui in Usingirr];
      Fu:=[pol/Coefficients(pol)[1] : pol in Fu];
      Z:=[Scheme(Pm,pol) : pol in Fu];
    end if;
  end if;
 end if;
end if;
```

Therefore, if one types the following code

```
> k:=29;  // for example
> load 'singular-varieties.m';
```

the program returns:

- `L`: the field over which the group G_k is defined in `Magma`: it is the smallest cyclotomic field over which G_k is defined.
- `K`: the field generated by traces of elements of G_k: it is the smallest field over which G_k is defined
- `W`: the group G_k, defined as a matrix group over `K`.
- `n`: the rank of the group
- `m`: the integer $n-1$
- `R`: the invariant ring $K[x_1,\dots,x_n]^{G_k}$
- `P`: the ring $K[x_1,\dots,x_n]$. If $n=3$ (resp. $n=4$, resp. $n \geqslant 5$), then the variables are called `x,y,z` (resp. `x,y,z,t`, resp. `x[1],x[2],...,x[n]`).
- `Pm`: the projective space $\mathbf{P}^m$, defined as a scheme over the field `K`.
- `dr, f1, f2,..., fr`: the natural number d_r and the fundamental invariants $f_1,\dots, f_r$. The program tests if the invariant `fr` of degree `dr` chosen with the command

  ```
  fr:=InvariantsOfDegree(W,dr)[1]
  ```

  is a fundamental invariant. It turns out that it is always the case, but it must be checked and the program then writes `fr is a undamental invariant`.
- `f`: the unique monomial f in $f_1,\dots, f_{r-1}$ of degree d_r.
- `AmxA1`: the affine space $\mathbf{A}^m \times \mathbf{A}^1$, defined as a scheme over the field `K`. Variables are denoted `u[1],u[2],...,u[n]`.
- `A1`: the affine space $\mathbf{A}^1$, defined as a scheme over the field `K`.
- `phi`: the map $\phi : \mathbf{A}^m \times \mathbf{A}^1 \longrightarrow \mathbf{A}^1$ given by the second projection.
- `affinization`: the map that sends an element f of $K[x_1,\dots,x_n]$ to the polynomial f^{aff}.
- `faff, fraff, Fuaff`: the polynomials f^{aff}, f_r^{aff} and F_u^{aff}.
- `X, Xsfib`: the varieties $\mathscr{X}$, $\mathscr{X}_{\mathrm{sfib}}$, defined as schemes over the field `K`.
- `Psing`: the unitary polynomial generating the defining ideal of $\phi(\mathscr{X}_{\mathrm{sfib}})$.
- `Using`: the roots of `Psing`. The programs checks if all roots belong to `K`: this is always the case, but it must be checked and the program then writes `All elements of U_sing belong to K`. So `Using` is the set U_{sing}.
- `Am`: the affine space $\mathbf{A}^m$, defined as a scheme over the field `K`.
- `Usingirr`: the set $U_{\mathrm{sing}}^{\mathrm{irr}}$. It must be noticed that, after the first command

  ```
  Usingirr:=[ui : ui in Using |
    IsIrreducible(Scheme(Am,Evaluate(Fuaff,x cat [ui])))];
  ```

  the set `Usingirr` is just the set of $u \in U_{\mathrm{sing}}$ such that $\mathscr{Z}(F_u^{\mathrm{aff}})$ is irreducible *as a scheme over* `K`. To check that it is irreducible over $\mathbb{C}$, we proceed as follows:
  - If $n=3$, then $\mathscr{Z}(F_u)$ is a curve and we define it *as a curve* in `Magma` using the command `Curve`, and we then use the command `IsAbsolutelyIrreducible`.
  - If $n \geqslant 4$, then we check that the dimension of the singular locus of $\mathscr{Z}(F_u^{\mathrm{aff}})$ is $\leqslant n-4$, so that $\mathscr{Z}(F_u^{\mathrm{aff}})$ is irreducible *over* $\mathbb{C}$ using Lemma 1.1.
- `Fu`: the list of polynomials F_{u_i} when u_i runs over $U_{\mathrm{sing}}^{\mathrm{irr}}$, after renormalization so that the first coefficient is 1.
- `Z, Zaff`: the list of varieties $\mathscr{Z}(F_{u_i})$ or $\mathscr{Z}^{\mathrm{aff}}(F_{u_i}^{\mathrm{aff}})$ when u_i runs over $U_{\mathrm{sing}}^{\mathrm{irr}}$.

2. Singular points

For computing singular points of the varieties $\mathscr{Z}(F_{u_i})$, we use the fact that all W-orbits of singular points meet the affine patch defined by $x_n \neq 0$. So this reduces the computations of singular points to an affine open subset, and helps to save much time. We then test the type of the singularity. The Magma code is contained in a file singular-points.m, and is shown below (note that, for saving much time in the computation whenever $W = G_{32}$, we use the fact that the varieties are defined over $\mathbb{Q}$ to make the computation of the reduced singular locus over $\mathbb{Q}$, and then we go back to the field K):

```
if k eq 32 then
  Z:=[ChangeRing(i,RationalField()) : i in Z];
end if;
if n eq 4 then
  Z:=[Surface(AmbientSpace(i),MinimalBasis(i)[1]) : i in Z];
end if;

print "Computing reduced singular loci...";
time Zsing:=[ReducedSubscheme(SingularSubscheme(AffinePatch(var,1))) : var in Z];
if k eq 32 then Z:=[ChangeRing(i,K) : i in Z];end if;
if k eq 32 then Zsing:=[ChangeRing(i,K) : i in Zsing];end if;

print "--------------------";
print "There are", # Z, "singular varieties of the form Z(F_u):";
print "they are of degre", dr;
for i in [1..# Z] do
  print "========================================";
  print "========================================";
  singularpoints:=Points(Zsing[i]);
  if # singularpoints eq Degree(Zsing[i]) then
    print "Singular points are defined over the coefficient ring";
    M:=CoefficientRing(Z[i]);
    ZM:=Z[i];
    WM:=W;
    singularpoints:=[ZM ! (Coordinates(k) cat [1]) : k in singularpoints];
  else
    print "Singular points are NOT defined over the coefficient ring";
    irr:=IrreducibleComponents(Zsing[i]);
    test:=Set([Degree(var) : var in irr]);
    while test ne {1} do
      liste:=[var : var in irr | Degree(var) gt 1];
      M:=CoordinateRing(liste[1]);
      irr:=[ChangeRing(var,M) : var in irr];
      irr:=[IrreducibleComponents(var) : var in irr];
      irr:= &cat irr;
      test:=Set([Degree(var) : var in irr]);
    end while;
    ZM:=ChangeRing(Z[i],M);
    WM:=ChangeRing(W,M);
    singularpoints:=[[ZM ! (Coordinates(k) cat [1]) : k in Points(var)] :
      var in irr];
    singularpoints:=&cat singularpoints;
  end if;
  orbits:=ProjectiveOrbits(WM,singularpoints);
  printf "Z(F_{u_%o})",i;
  print " has", &+ [k[2] : k in orbits], "singular points";
  if # orbits eq 1 then
    print "There is only 1 orbit of singular points";
  else
    print "There are", # orbits, "orbits of singular points";
```

```
  end if;
  for j in [1..# orbits] do
    print "--------";
    print "  -> Orbit number", j, "contains", orbits[j][2], "points";
    pt:=ZM ! Coordinates(orbits[j][1]);
    if IsNode(pt) then
      print "     They are all nodes";
    elif (n eq 3) and IsCusp(pt) then
      print "     They are all cusps";
    elif (n eq 3) then
      boo,pol,type:=HypersurfaceSingularity(ZM,pt);
      print "     They are all of type", type;
    elif n eq 4 and IsSimpleSurfaceSingularity(pt) then
      a,b,c:=IsSimpleSurfaceSingularity(pt);
      print "     They are all simple singularities of type", b, c;
    else
      if n eq 4 then
        print "     They are not simple singularities";
      end if;
      print "     IsOrdinary :",IsOrdinarySingularity(pt);
      print "     multiplicity =",Multiplicity(pt);
      print "     Milnor number =",HypersurfaceMilnorNumber(ZM,pt);
      print "     Tjurina number =",HypersurfaceTjurinaNumber(ZM,pt);
    end if;
  end for;
  print "  ";
end for;
```

For instance, if one wants to check the results of [**Bon2**, Table 4.2] for the group G_{29}, it is sufficient to download the previous files and type:

```
k:=29; // for instance
load 'singular-varieties.m';
load 'singular-points.m';
```

The results are given case-by-case in the next sections.

3. Rank 3

Once the computation of $U^{\mathrm{irr}}_{\mathrm{sing}}$ has been finished, desribing the orbits of singular points is almost immediate in the cases treated in this section. The results of the computation described in this section prove [**Bon2**, Proposition 3.1 for groups of rank 3] and [**Bon2**, Table 3.2].

3.A. The group $G_{23} = \mathrm{W}(H_3)$. —

```
> k:=23;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 0.020
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 0.180
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 0.010
--------------------
There are 2 singular varieties of the form Z(F_u):
they are of degree 6
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 6 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 6 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 10 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 10 points
     They are all nodes

> Cputime();
0.690
```

3.B. The group G_{24}. —

```
> k:=24;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 1.580
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 1094.150
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 0.220
--------------------
There are 3 singular varieties of the form Z(F_u):
they are of degre 14
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 21 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 21 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 28 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 28 points
     They are all nodes

========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_3}) has 42 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 42 points
     They are all cusps

> Cputime();
1097.890
```

Let us give the equation of the curve $\mathscr{Z}(F_{u_3})$ with 42 cusps: one can check that all singular points live in the open subset $z \neq 0$, so that the equation of the curve might be written in $\mathbb{C}[x,y]$ (and in fact in $\mathbb{Q}[x,y]$):

```
x^14 - 1/5*x^12*y^2 - 7/10*x^12*y + 7/40*x^12 - 201/35*x^10*y^4
  + 21/5*x^10*y^3 - 81/40*x^10*y^2 + 483/80*x^10*y + 525/128*x^10
  - 437/245*x^8*y^6 - 159/70*x^8*y^5 + 5133/280*x^8*y^4
  + 67/10*x^8*y^3 - 93/320*x^8*y^2 + 3927/640*x^8*y - 3997/2560*x^8
  + 3033/1715*x^6*y^8 + 6/245*x^6*y^7 + 8529/980*x^6*y^6 + 207/40*x^6*y^5
  + 5997/2240*x^6*y^4 - 4029/320*x^6*y^3 + 6723/1280*x^6*y^2 + 21/10*x^6*y
  + 1491/2048*x^6 - 407/2401*x^4*y^10 + 29/686*x^4*y^9 + 27423/13720*x^4*y^8
  - 1437/490*x^4*y^7 - 876/245*x^4*y^6 - 153/1120*x^4*y^5 - 9357/4480*x^4*y^4
  - 3/2*x^4*y^3 + 657/5120*x^4*y^2 - 5033/10240*x^4*y + 5509/40960*x^4
  + 6557/84035*x^2*y^12 + 213/12005*x^2*y^11 + 18471/96040*x^2*y^10
  - 12223/27440*x^2*y^9 + 217143/219520*x^2*y^8 - 153/448*x^2*y^7
  + 14349/62720*x^2*y^6 + 9/70*x^2*y^5 + 50553/71680*x^2*y^4
  + 719/5120*x^2*y^3 - 2361/20480*x^2*y^2 - 1491/10240*x^2*y - 371/40960*x^2
  + 1119/4117715*y^14 + 99/33614*y^13 + 6987/672280*y^12 + 213/24010*y^11
  - 21087/768320*y^10 - 18717/219520*y^9 - 66033/878080*y^8 + 27/490*y^7
  - 279/7840*y^6 + 2157/71680*y^5 + 7083/286720*y^4 - 213/10240*y^3
  + 159/40960*y^2 - 3/81920
```

3.C. The group G_{25}. —

```
> k:=25;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 0.050
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 2.420
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 0.010
--------------------
There are 2 singular varieties of the form Z(F_u):
they are of degre 12
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 12 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 12 points
     They are all of type D4

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 36 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 36 points
     They are all cusps

> Cputime();
3.170
```

3.D. The group G_{27}. —

```
> k:=27;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 0.370
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 38.170
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 0.050
--------------------
There are 2 singular varieties of the form Z(F_u):
they are of degre 12
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 45 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 45 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 36 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 36 points
     They are all nodes

> Cputime();
48.260
```

4. The group G_{26}

Assume in this section, and only in this section, that $W = G_{26}$. As explained in [**Bon2**, Remark 3.4], the problem of finding singular curves of the form $\mathscr{Z}(F_u)$ for G_{26} is the same problem as for G_{25}. However, as explained in [**Bon2**, Example 3.5], it is possible to construct singular curves of degree 18 from invariants of G_{26} as follows. Any fundamental invariant of degree 18 is, up to scalar, of the form $F_{u,v} = f_3 + u f_1 f_2 + v f_1^3$. One can determine the set $\mathscr{C}$ of $(u,v) \in \mathbb{C}^2$ such that $\mathscr{Z}(F_{u,v})$ is singular ($\mathscr{C}$ has dimension 1) and determine its reduced singular locus $\mathscr{C}_{\text{sing}}$ (which is a union of 6 points). Then we study the singularities of $\mathscr{Z}(F_{u,v})$ for all $(u,v) \in \mathscr{C}_{\text{sing}}$ in the same way as for other groups of rank 3 (one must modify slightly the printing of the results from the file `singularpoints.m`). The code is contained in the file `g26-degree18.m` below:

```
load 'auxiliary-functions.m';
load 'primitive-complex-reflection-groups.m';
W:=PrimitiveComplexReflectionGroup(26);
K<zeta>:=CoefficientRing(W);
R:=InvariantRing(W);
P<x,y,z>:=PolynomialRing(R);
P2:=Proj(P);

f1:=InvariantsOfDegree(W,6)[1];
f2:=InvariantsOfDegree(W,12)[1];
f3:=InvariantsOfDegree(W,18)[1];

if Gcd(f1,f2) eq 1 then print "f2 is a fundamental invariant";end if;
if Gcd(f1,f3) eq 1 then print "f3 is a fundamental invariant";end if;

A2xA2<X,Y,u,v>:=AffineSpace(K,4);
A2:=AffineSpace(K,2);
phi:=map<A2xA2->A2 | [u,v]>;
affinization:=function(p) return Evaluate(p,[X,Y,1]);end function;
f1aff:=affinization(f1);
f2aff:=affinization(f2);
f3aff:=affinization(f3);
Fuvaff:=f3aff + u*f2aff*f1aff + v * f1aff^3;

X:=Scheme(A2xA2,Fuvaff);
Xsfib:=Scheme(X,[Derivative(Fuvaff,i) : i in [1,2]]);
print "Computing phi(Xsfib)...";
time C:=phi(Xsfib);
C:=ReducedSubscheme(C);
Csing:=SingularSubscheme(C);
Csing:=ReducedSubscheme(Csing);
csing:=SingularPoints(C);
if # csing eq Degree(Csing) then
  print "Singular parameters are defined over K";
end if;
print "Checking absolute irreducibility...";
time csingirr:=[pt : pt in csing |
        IsAbsolutelyIrreducible(Curve(P2,f3 + pt[1]*f2*f1 + pt[2] * f1^3))];

c:=# csingirr;
Z:=[Curve(P2,f3 + pt[1]*f2*f1 + pt[2] * f1^3) : pt in csingirr];
print "Computing reduced singular loci...";
Zsing:=[ReducedSubscheme(SingularSubscheme(AffinePatch(var,1))) : var in Z];
```

```
print "--------------------";
print "There are", # Z, "singular varieties of the form Z(F_u):";
print "they are of degre 18";
for i in [1..# Z] do
  print "========================================";
  print "========================================";
  singularpoints:=Points(Zsing[i]);
  if # singularpoints eq Degree(Zsing[i]) then
    print "Singular points are defined over the coefficient ring";
    M:=CoefficientRing(Z[i]);
    ZM:=Z[i];
    WM:=W;
    singularpoints:=[ZM ! (Coordinates(k) cat [1]) : k in singularpoints];
  else
    print "Singular points are NOT defined over the coefficient ring";
    irr:=IrreducibleComponents(Zsing[i]);
    test:=Set([Degree(var) : var in irr]);
    while test ne {1} do
      liste:=[var : var in irr | Degree(var) gt 1];
      M:=CoordinateRing(liste[1]);
      irr:=[ChangeRing(var,M) : var in irr];
      irr:=[IrreducibleComponents(var) : var in irr];
      irr:= &cat irr;
      test:=Set([Degree(var) : var in irr]);
    end while;
    ZM:=ChangeRing(Z[i],M);
    WM:=ChangeRing(W,M);
    singularpoints:=[[ZM ! (Coordinates(k) cat [1]) : k in Points(var)] : var in irr];
    singularpoints:=&cat singularpoints;
  end if;
  orbits:=ProjectiveOrbits(WM,singularpoints);
  printf "Z(F_{u_%o,v_%o})",i,i;
  print " has", &+ [k[2] : k in orbits], "singular points";
  if # orbits eq 1 then
    print "There is only 1 orbit of singular points";
  else
    print "There are", # orbits, "orbits of singular points";
  end if;
  for j in [1..# orbits] do
    print "--------";
    print "  -> Orbit number", j, "contains", orbits[j][2], "points";
    pt:=ZM ! Coordinates(orbits[j][1]);
    if IsNode(pt) then
      print "     They are all nodes";
    elif IsCusp(pt) then
      print "     They are all cusps";
    else
      boo,pol,type:=HypersurfaceSingularity(ZM,pt);
      print "     They are all of type", type;
    end if;
  end for;
  print "  ";
end for;
```

The result is given below (loading the whole file takes about 4 minutes). It proves [**Bon2**, Table 3.6].

```
> load 'g26-degree18.m';
Loading "g26-degree18.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
f2 is a fundamental invariant
f3 is a fundamental invariant
Computing phi(Xsfib)...
Time: 47.580
Singular parameters are defined over K
Checking absolute irreducibility...
Time: 145.930
Computing reduced singular loci...
Time: 0.080
--------------------
There are 5 singular varieties of the form Z(F_u):
they are of degre 18
========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_1,v_1}) has 63 singular points
There are 2 orbits of singular points
--------
  -> Orbit number 1 contains 9 points
     They are all of type X9
--------
  -> Orbit number 2 contains 54 points
     They are all cusps

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2,v_2}) has 21 singular points
There are 2 orbits of singular points
--------
  -> Orbit number 1 contains 12 points
     They are all of type D4
--------
  -> Orbit number 2 contains 9 points
     They are all of type X9

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_3,v_3}) has 45 singular points
There are 2 orbits of singular points
--------
  -> Orbit number 1 contains 9 points
     They are all of type X9
--------
  -> Orbit number 2 contains 36 points
     They are all cusps

========================================
```

```
========================================
Singular points are defined over the coefficient ring
Z(F_{u_4,v_4}) has 36 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 36 points
     They are all of type E6

========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_5,v_5}) has 84 singular points
There are 2 orbits of singular points
--------
  -> Orbit number 1 contains 12 points
     They are all of type D4
--------
  -> Orbit number 2 contains 72 points
     They are all cusps

> Cputime();
228.540
```

The curve with 36 singular points of type E_6 is given by the following polynomial (restricted to the open affine subset defined by $z \neq 0$):

```
x^18 + 2*x^15*y^3 + 2*x^15 - 17*x^12*y^6 - 262*x^12*y^3 - 17*x^12
  + 28*x^9*y^9 - 1788*x^9*y^6 - 1788*x^9*y^3 + 28*x^9 - 17*x^6*y^12
  - 1788*x^6*y^9 - 8166*x^6*y^6 - 1788*x^6*y^3 - 17*x^6 + 2*x^3*y^15
  - 262*x^3*y^12 - 1788*x^3*y^9 - 1788*x^3*y^6 - 262*x^3*y^3 + 2*x^3
  + y^18 + 2*y^15 - 17*y^12 + 28*y^9 - 17*y^6 + 2*y^3 + 1
```

5. Rank 4

The results of the computation described in this section prove [**Bon2**, Proposition 3.1 for groups of rank 4] and [**Bon2**, Table 4.2].

5.A. The group $G_{28} = W(F_4)$. — In this case, the results were already obtained by Sarti [**Sar**].

```
> k:=28;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 0.110
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 0.050
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 0.050
--------------------
There are 4 singular varieties of the form Z(F_u):
they are of degree 6
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 12 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 12 points
     They are all nodes


========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 12 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 12 points
     They are all nodes


========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_3}) has 48 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 48 points
     They are all nodes

========================================
========================================
```

```
Singular points are defined over the coefficient ring
Z(F_{u_4}) has 48 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 48 points
     They are all nodes

> Cputime();
0.630
```

5.B. The group G_{29}. —

```
> k:=29;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 1.000
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 0.110
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 0.230
--------------------
There are 5 singular varieties of the form Z(F_u):
they are of degre 8
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 40 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 40 points
     They are all nodes


========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 20 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 20 points
     They are not simple singularities
     IsOrdinary : true
     multiplicity = 3
     Milnor number = 11
     Tjurina number = 10


========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_3}) has 160 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 160 points
     They are all nodes


========================================
========================================
Singular points are defined over the coefficient ring
```

```
Z(F_{u_4}) has 80 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 80 points
     They are all nodes


========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_5}) has 80 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 80 points
     They are all nodes

> Cputime();
7.970
```

5.C. The group $G_{30} = W(H_4)$. — In this case, the results were already obtained by Sarti [**Sar**].

```
> k:=30;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 21.810
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 1.380
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 23.630
--------------------
There are 4 singular varieties of the form Z(F_u):
they are of degre 12
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 60 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 60 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 300 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 300 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_3}) has 360 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 360 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_4}) has 600 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 600 points
     They are all nodes

> Cputime();
```

```
255.150
```

5.D. The group G_{31}. — The computation below proves [**Bon2**, Table 4.2] in this case.

```
> k:=31;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 804.310
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 7.990
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 31.190
--------------------
There are 5 singular varieties of the form Z(F_u):
they are of degre 20
========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_1}) has 640 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 640 points
     They are all nodes

========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_2}) has 1920 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 1920 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_3}) has 480 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 480 points
     They are all nodes

========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_4}) has 1440 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 1440 points
     They are all nodes

========================================
```

```
========================================
Singular points are defined over the coefficient ring
Z(F_{u_5}) has 960 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 960 points
     They are all nodes

> Cputime();
1147.150
```

5.E. The group G_{32}. —

```
> k:=32;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 245.490
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 6.770
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 418.470
--------------------
There are 4 singular varieties of the form Z(F_u):
they are of degre 24
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 40 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 40 points
     They are not simple singularities
     IsOrdinary : true
     multiplicity = 6
     Milnor number = 125
     Tjurina number = 125

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 360 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 360 points
     They are not simple singularities
     IsOrdinary : false
     multiplicity = 3
     Milnor number = 18
     Tjurina number = 18

========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_3}) has 1440 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 1440 points
     They are all simple singularities of type D 4

========================================
========================================
```

```
Singular points are NOT defined over the coefficient ring
Z(F_{u_4}) has 540 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 540 points
     They are not simple singularities
     IsOrdinary : false
     multiplicity = 2

> Cputime();
2804.840
```

6. Rank $\geqslant 5$

6.A. The group G_{33}. — The computation below proves [**Bon2**, Example 5.1].

```
> k:=33;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 2662.980
There is no singular variety of the form Z(F_u)
```

6.B. The group $G_{35} = W(E_6)$. — The computation below proves [**Bon2**, Example 5.2].

```
> k:=35;
> load 'singular-varieties.m';
Loading "singular-varieties.m"
Loading "auxiliary-functions.m"
Loading "primitive-complex-reflection-groups.m"
fr is a fundamental invariant
Computing phi(Xsfib)...
Time: 2048.600
All elements of U_sing belong to K
Checking absolute irreducibility...
Time: 79.470
> load 'singular-points.m';
Loading "singular-points.m"
Computing reduced singular loci...
Time: 1171.100
--------------------
There are 8 singular varieties of the form Z(F_u):
they are of degre 6
========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_1}) has 36 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 36 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_2}) has 360 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 360 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_3}) has 27 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 27 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_4}) has 135 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 135 points
     They are all nodes
========================================
========================================
```

```
Singular points are NOT defined over the coefficient ring
Z(F_{u_5}) has 432 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 432 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_6}) has 1080 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 1080 points
     They are all nodes

========================================
========================================
Singular points are NOT defined over the coefficient ring
Z(F_{u_7}) has 1080 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 1080 points
     They are all nodes

========================================
========================================
Singular points are defined over the coefficient ring
Z(F_{u_8}) has 216 singular points
There is only 1 orbit of singular points
--------
  -> Orbit number 1 contains 216 points
     They are all nodes

> Cputime();
3508.070
```

7. Some equations

7.A. Equation of Sarti surface over $\mathbb{Q}$. — According to the proof of [**Bon2**, Prop. 1.1], we must start with a model of $G_{30} = \mathrm{W}(H_4)$ which is Galois invariant. Such a model is of course non-unique, and we have chosen the following one:

```
K<zeta>:=CyclotomicField(20);
z:=zeta^4;
i:=zeta^5;
r5:=(2*z^3 + 2*z^2 + 1);  // sqrt(5)
K<r>:=sub<K | i*(z^3-z^2)>;  // r^4-5*r^2+5 = 0
auto:=Automorphisms(K);
PK<[a]>:=PolynomialRing(K,4);
A3:=Spec(PK);

POL<X>:=PolynomialRing(K);

s1:=Matrix(K,4,4,
[[-1,0,0,0],
 [ 0,1,0,0],
 [ 0,0,1,0],
 [ 0,0,0,1]]);

s2:=Matrix(K,4,4,
[[-1/2*(r^2-3), r/2,0,0],
 [r/2, 1/2*(r^2-3),0,0],
 [0,0,1,0],
 [0,0,0,1]]);

s3:=Matrix(K,4,4,
[[1,0,0,0],
 [0,-1/5*(2*r^2 - 5),2/r5,0],
 [0,2/r5,1/5*(2*r^2 - 5),0],
 [0,0,0,1]]);

s4:=Matrix(K,4,4,
[1,0,0,0,
 0,1,0,0,
 0,0,-1/2*(r^2 - 2),1/2*(r^3-3*r),
 0,0,1/2*(r^3- 3*r),1/2*(r^2 - 2)]);

W:=MatrixGroup<4,K | s1,s2,s3,s4>;  // This is W(H_4)
```

We then slightly modify the code of `singular-varities.m` to obtain the varieties $\mathscr{Z}(F_u)$, $u \in U_{\mathrm{sing}}^{\mathrm{irr}}$. The computation is much faster than with the other model: about 4 minutes instead of 21. Astonishingly, most of the time is spent to convert the scheme $\mathscr{Z}(F_u)$ into the type `Srf` (surface) of `Magma`. Since we know *a priori* that points are nodes, this could be avoided. This leads to the following polynomial (at least its restriction to the open subset $t \neq 0$):

```
x^12 - 177/11*x^10*y^2 - 177/11*x^10*z^2 - 177/11*x^10 + 894/11*x^8*y^4
  + 1788/11*x^8*y^2*z^2 + 1788/11*x^8*y^2 + 894/11*x^8*z^4 + 1788/11*x^8*z^2
  + 894/11*x^8 + 614/55*x^6*y^6 - 12462/11*x^6*y^4*z^2 - 12462/11*x^6*y^4
  + 3576/11*x^6*y^2*z^4 + 7152/11*x^6*y^2*z^2 + 3576/11*x^6*y^2
  - 1458/5*x^6*y*z^5 + 2916*x^6*y*z^3 - 1458*x^6*y*z - 1481/11*x^6*z^6
  - 4443/11*x^6*z^4 - 4443/11*x^6*z^2 - 1481/11*x^6 - 3549/55*x^4*y^8
  + 49956/55*x^4*y^6*z^2 + 49956/55*x^4*y^6 + 5364/11*x^4*y^4*z^4
  + 10728/11*x^4*y^4*z^2 + 5364/11*x^4*y^4 + 1458/5*x^4*y^3*z^5
  - 2916*x^4*y^3*z^3 + 1458*x^4*y^3*z - 4443/11*x^4*y^2*z^6
  - 13329/11*x^4*y^2*z^4 - 13329/11*x^4*y^2*z^2 - 4443/11*x^4*y^2
  + 1458/5*x^4*y*z^7 - 13122/5*x^4*y*z^5 - 1458*x^4*y*z^3 + 1458*x^4*y*z
  + 894/11*x^4*z^8 + 3576/11*x^4*z^6 + 5364/11*x^4*z^4 + 3576/11*x^4*z^2
  + 894/11*x^4 + 6267/275*x^2*y^10 - 15117/55*x^2*y^8*z^2 - 15117/55*x^2*y^8
  + 3576/11*x^2*y^6*z^4 + 7152/11*x^2*y^6*z^2 + 3576/11*x^2*y^6
  + 13122/25*x^2*y^5*z^5 - 26244/5*x^2*y^5*z^3 + 13122/5*x^2*y^5*z
  - 4443/11*x^2*y^4*z^6 - 13329/11*x^2*y^4*z^4 - 13329/11*x^2*y^4*z^2
  - 4443/11*x^2*y^4 - 2916/5*x^2*y^3*z^7 + 26244/5*x^2*y^3*z^5
  + 2916*x^2*y^3*z^3 - 2916*x^2*y^3*z + 1788/11*x^2*y^2*z^8
  + 7152/11*x^2*y^2*z^6 + 10728/11*x^2*y^2*z^4 + 7152/11*x^2*y^2*z^2
  + 1788/11*x^2*y^2 - 1752/275*x^2*z^10 - 15117/55*x^2*z^8 + 49956/55*x^2*z^6
  - 12462/11*x^2*z^4 + 1788/11*x^2*z^2 - 177/11*x^2 + 32/275*y^12
  - 1752/275*y^10*z^2 - 1752/275*y^10 + 894/11*y^8*z^4 + 1788/11*y^8*z^2
  + 894/11*y^8 - 1458/25*y^7*z^5 + 2916/5*y^7*z^3 - 1458/5*y^7*z - 1481/11*y^6*z^6
  - 4443/11*y^6*z^4 - 4443/11*y^6*z^2 - 1481/11*y^6 + 1458/25*y^5*z^7
  - 13122/25*y^5*z^5 - 1458/5*y^5*z^3 + 1458/5*y^5*z + 894/11*y^4*z^8
  + 3576/11*y^4*z^6 + 5364/11*y^4*z^4 + 3576/11*y^4*z^2 + 894/11*y^4
  - 1752/275*y^2*z^10 - 15117/55*y^2*z^8 + 49956/55*y^2*z^6 - 12462/11*y^2*z^4
  + 1788/11*y^2*z^2 - 177/11*y^2 + 32/275*z^12 + 6267/275*z^10
  - 3549/55*z^8 + 614/55*z^6 + 894/11*z^4 - 177/11*z^2 + 1
```

Even though this polynomial looks quite ugly (and might possibly be transformed to a nicer one by clever change of coordinates), the interested reader can copy this polynomial and plug it in the software `SURFER`: he will recognize the Sarti surface.

8. Complements

8.A. Computations for the octic with 48 singularities of type D_4. — The computation is almost immediate, even on a standard computer.

```
> load 'auxiliary-functions.m';
Loading "auxiliary-functions.m"
> load 'primitive-complex-reflection-groups.m';
Loading "primitive-complex-reflection-groups.m"
>
> W:=PrimitiveComplexReflectionGroup(28);
> R:=InvariantRing(W);
> P<x,y,z,t>:=PolynomialRing(R);
> P3:=Proj(P);
> s1:=ElementarySymmetricPolynomial(P,1);
> s2:=ElementarySymmetricPolynomial(P,2);
```

```
> s3:=ElementarySymmetricPolynomial(P,3);
> s4:=ElementarySymmetricPolynomial(P,4);
> c:=function(pol) return Evaluate(pol,[x^2,y^2,z^2,t^2]); end function;
>
> pol:=7*c(s1)^4-72*c(s1)^2*c(s2)+4320*c(s4)+432*c(c(s2));
> pol in R;
true
> var:=Surface(P3,pol);
> varsing:=ReducedSubscheme(SingularSubscheme(var));
> Degree(varsing);
48
> singularpoints:=SingularPoints(var);
> # singularpoints;
48
> # ProjectiveOrbits(W,singularpoints);
1
> IsSimpleSurfaceSingularity(var ! singularpoints[1]);
true D 4
```

8.B. Computations for the dodecic with 160 singularities of type D_4. — Only the computation of the reduced singular locus takes some time (more than 67 minutes):

```
> load 'auxiliary-functions.m';
Loading "auxiliary-functions.m"
> load 'primitive-complex-reflection-groups.m';
Loading "primitive-complex-reflection-groups.m"
>
> W:=PrimitiveComplexReflectionGroup(29);
> R:=InvariantRing(W);
> P<x,y,z,t>:=PolynomialRing(R);
> P3:=Proj(P);
> s1:=ElementarySymmetricPolynomial(P,1);
> s2:=ElementarySymmetricPolynomial(P,2);
> s3:=ElementarySymmetricPolynomial(P,3);
> s4:=ElementarySymmetricPolynomial(P,4);
> c:=function(pol) return Evaluate(pol,[x^2,y^2,z^2,t^2]); end function;
>
> pol:=2*c(s1)^6-3*c(s1)^4*c(s2)-156*c(s1)^2*c(s2)^2 +585*c(s1)^3*c(s3)+416*c(s2)^3
> -1980*c(s1)*c(s2)*c(s3) +3420 *c(s1)^2*c(s4)+2700*c(s3)^2-5760*c(s2)*c(s4);
> pol in R;
true
> var:=Surface(P3,pol);
> time varsing:=ReducedSubscheme(SingularSubscheme(var));
Time: 4041.200
> Degree(varsing);
160
> singularpoints:=SingularPoints(var);
> # singularpoints;
160
> # ProjectiveOrbits(W,singularpoints);
1
> IsSimpleSurfaceSingularity(var ! singularpoints[1]);
true D 4
```

Appendix I: Primitive complex reflection groups as defined in CHAMP

Below is the file primitive-complex-reflection-groups.m:

```
PrimitiveComplexReflectionGroup:=function(i) local W; // i in {4,5,6,...,37}
  // G4
  if i eq 4 then
    W:=MatrixGroup<2, K |
    [DiagonalMatrix(K, [ 1, w ]),
     Matrix(K, 2, 2,[ [1/3*(2*w + 1), 1/3*(2*w - 2)],
                    [  1/3*(w - 1),   1/3*(w + 2)] ])] >
                    where w := K.1 where K := CyclotomicField(3);
  // G5
  elif i eq 5 then
    W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, w ]),
                           Matrix(K, 2, 2, [ [ 2/3, 1/3 ], [ 2/3, -2/3 ],
                           [ 1/3, -1/3 ], [ 1/3, 2/3 ] ])]>
                           where w := K.1 where K := CyclotomicField(3);
  // G6
  elif i eq 6 then
    W:=MatrixGroup<2, K | [ Matrix(K, 2, 2,[ [ 0, 2/3, 0, -1/3 ], [ 0, 4/3, 0, -2/3 ],
    [ 0, 2/3, 0, -1/3 ], [ 0, -2/3, 0, 1/3 ] ]),
                            DiagonalMatrix(K, [ 1, w^2 - 1 ])]>
                            where w := K.1 where K := CyclotomicField(12);
  // G7
  elif i eq 7 then
    W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, -1 ]),
                           Matrix(K, 2, 2,[ [ 0, 1/2, 1/2, -1/2 ], [ 0, 1/2, -1/2, -1/2 ],
                           [ 0, 1/2, 1/2, -1/2 ], [ 0, -1/2, 1/2, 1/2 ] ]),
                           Matrix(K, 2, 2,[ [ 0, 1/2, 1/2, -1/2 ],
                           [ 0, 1/2, 1/2, -1/2 ], [ 0, 1/2, -1/2, -1/2 ], [ 0, -1/2, 1/2, 1/2 ] ])]>
                           where w := K.1 where K := CyclotomicField(12);
  // G8
  elif i eq 8 then
    W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, w ]),
                           Matrix(K, 2, 2,[ [ 1/2, 1/2 ], [ -1/2, 1/2 ],
                           [ -1/2, 1/2 ], [ 1/2, 1/2 ] ])]>
                           where w := K.1 where K := CyclotomicField(4);
  // G9
  elif i eq 9 then
    W:=MatrixGroup<2, K | [Matrix(K, 2, 2,[ [ 0, 1/2, 0, -1/2 ], [ 1/2, 0, 0, 0 ],
    [ 1, 0, 0, 0 ], [ 0, -1/2, 0, 1/2 ] ]),
                           DiagonalMatrix(K, [ 1, w^2 ])]>
                           where w := K.1 where K := CyclotomicField(8);
  // G10
  elif i eq 10 then
    W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, w^2 - 1 ]),
                           Matrix(K,2,2,[[2/3, 1/3, -1/3, 1/3 ],[-1/3, -2/3, 2/3, 1/3 ],
                           [-1/6, -1/3, 1/3, 1/6 ],[1/3, -1/3, 1/3, 2/3 ]])]>
                           where w := K.1 where K := CyclotomicField(12);
  // G11
  elif i eq 11 then
    W:=MatrixGroup<2, K | [Matrix(K, 2, 2,[ [ 0, -1/3, 0, -1/3, 0, -1/3, 0, 2/3 ],
                          [ 0, -1/3, 0, 2/3, 0, 2/3, 0, -1/3 ],
                           [ 0, 1/3, 0, -1/6, 0, -1/6, 0, -1/6 ], [ 0, 1/3, 0, 1/3, 0, 1/3, 0, -2/3 ] ]),
                           DiagonalMatrix(K, [ 1, w^4 - 1 ]),
                           Matrix(K, 2, 2,[ [ 2/3, 0, 1/3, 0, -1/3, 0, 1/3, 0 ],
                           [ -1/3, 0, -2/3, 0, 2/3, 0, 1/3, 0 ],
                                            [ -1/6, 0, -1/3, 0, 1/3, 0, 1/6, 0 ],
                                            [ 1/3, 0, -1/3, 0, 1/3, 0, 2/3, 0 ] ])]>
                           where w := K.1 where K := CyclotomicField(24);
  // G12
  elif i eq 12 then
    W:=MatrixGroup<2, K | [Matrix(K, 2, 2,[ [ 1/2, 0, 0, 0 ], [ 1, 1/2, 0, 1/2 ],
                          [ 1/2, -1/4, 0, -1/4 ], [ -1/2, 0, 0, 0 ] ]),
                           Matrix(K, 2, 2,[ [ 1/2, 0, 0, 0 ], [ 1, -1/2, 0, -1/2 ],
                           [ 1/2, 1/4, 0, 1/4 ], [ -1/2, 0, 0, 0 ] ]),
                           DiagonalMatrix(K, [ 1, -1 ])]>
                           where w := K.1 where K := CyclotomicField(8);
  // G13
  elif i eq 13 then
    W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, -1 ]),
                           Matrix(K, 2, 2,[ [ 0, 1/2, 0, -1/2 ], [ 0, -1/2, 0, 1/2 ],
                           [ 0, -1/2, 0, 1/2 ], [ 0, -1/2, 0, 1/2 ] ]),
                           Matrix(K, 2, 2,[ [ 0, 1/2, 0, -1/2 ], [ 0, 1/2, 0, 1/2 ],
                           [ 0, -1/2, 0, -1/2 ], [ 0, -1/2, 0, 1/2 ] ])]>
                           where w := K.1 where K := CyclotomicField(8);
  // G14
  elif i eq 14 then
    W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, -1 ]),
                           Matrix(K, 2, 2,[ [ 0, 1/2, 0, 0, 1/2, 0, 0, -1/2 ],
                           [ 0, 0, 0, 0, 1/2, 0, 0, 0 ],
                                            [ 0, 0, 0, 0, -1/2, 0, 0, 0 ],
                                            [ 0, -1/2, 0, 0, 1/2, 0, 0, 1/2 ] ])]>
                           where w := K.1 where K := CyclotomicField(24);
```

```
// G15
elif i eq 15 then
  W:=MatrixGroup<2, K | [Matrix(K, 2, 2,[ [ 0, -1/3, 0, -1/3, 0, -1/3, 0, 2/3 ],
                         [ 0, -1/3, 0, 2/3, 0, 2/3, 0, -1/3 ],
                                          [ 0, 1/3, 0, -1/6, 0, -1/6, 0, -1/6 ],
                                          [ 0, 1/3, 0, 1/3, 0, 1/3, 0, -2/3 ] ]),
                         DiagonalMatrix(K, [ 1, w^4 - 1 ]),
                         Matrix(K, 2, 2,[ [ 0, 0, 2/3, 0, 0, 0, -1/3, 0 ], [ 0, 0, -4/3, 0, 0, 0, 2/3, 0 ],
                                          [ 0, 0, -2/3, 0, 0, 0, 1/3, 0 ], [ 0, 0, -2/3, 0, 0, 0, 1/3, 0 ] ])]>
                         where w := K.1 where K := CyclotomicField(24);
// G16
elif i eq 16 then
  W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, w ]),
                         Matrix(K, 2, 2,[ [ 1/5, 2/5, -2/5, -1/5 ], [ -2/5, 1/5, -1/5, 2/5 ],
                                          [ -2/5, 1/5, -1/5, -3/5 ], [ 4/5, 3/5, 2/5, 1/5 ] ])]>
                         where w := K.1 where K := CyclotomicField(5);
// G17
elif i eq 17 then
  W:=MatrixGroup<2, K | [Matrix(K, 2, 2,[ [ 0, 2/5, 0, 1/5, 0, 1/5, 0, -3/5 ],
                        [ 0, -4/5, 0, 3/5, 0, -2/5, 0, 1/5 ],
                                          [ 0, -4/5, 0, 3/5, 0, -2/5, 0, 1/5 ],
                                          [ 0, -2/5, 0, -1/5, 0, -1/5, 0, 3/5 ] ]),
                         DiagonalMatrix(K, [ 1, w^4 ])]>
                         where w := K.1 where K := CyclotomicField(20);
// G18
elif i eq 18 then
  W:=MatrixGroup<2, K | [Matrix(K, 2, 2,[ [ 4/5, -1/5, 0, 0, -2/5, 4/5, 0, -3/5 ],
                        [ 2/5, -3/5, 0, 0, -1/5, 2/5, 0, 1/5 ],
                                          [ -3/5, 2/5, 0, 0, 4/5, -3/5, 0, 1/5 ],
                                          [ 1/5, 1/5, 0, 0, 2/5, 1/5, 0, 3/5 ] ]),
                         DiagonalMatrix(K, [ 1, w^3 ])]>
                         where w := K.1 where K := CyclotomicField(15);
// G19
elif i eq 19 then
  W:=MatrixGroup<2, K |
  [Matrix(K, 2, 2,[ [ 0, 3/5, 0, 2/5, 0, 0, 0, 0, 0, 1/5, 0, -3/5, 0, 0, 0, 1/5 ],
                    [ 0, -1/5, 0, -4/5, 0, 0, 0, 0, 0, 3/5, 0, 1/5, 0, 0, 0, -2/5 ],
                    [ 0, -1/5, 0, -4/5, 0, 0, 0, 0, 0, 3/5, 0, 1/5, 0, 0, 0, -2/5 ],
                    [ 0, -3/5, 0, -2/5, 0, 0, 0, 0, 0, -1/5, 0, 3/5, 0, 0, 0, -1/5 ] ]),
   Matrix(K, 2, 2,[ [ -1/5, 0, 2/5, 0, 2/5, 0, 1/5, 0, 1/5, 0, 2/5, 0, -3/5, 0, -1/5, 0 ],
                    [ 2/5, 0, 1/5, 0, -4/5, 0, -2/5, 0, -2/5, 0, 1/5, 0, 1/5, 0, 2/5, 0 ],
                    [ -3/5, 0, -4/5, 0, 1/5, 0, 3/5, 0, 3/5, 0, 1/5, 0, 1/5, 0, -3/5, 0 ],
                    [ 1/5, 0, -2/5, 0, -2/5, 0, -1/5, 0, -1/5, 0, 3/5, 0, 3/5, 0, 1/5, 0 ] ]),
                         DiagonalMatrix(K, [ 1, w^12 ])]>
                         where w := K.1 where K := CyclotomicField(60);
// G20
elif i eq 20 then
  W:=MatrixGroup<2, K | [DiagonalMatrix(K, [ 1, w^5 ]),
                         Matrix(K, 2, 2,[ [ 1, -1/3, -2/3, 2/3, -1/3, 2/3, 0, -2/3 ],
                         [ -1, 2/3, 4/3, -4/3, 2/3, -1/3, 0, 4/3 ],
                                          [ -1/5, 2/15, 4/15, -4/15, 2/15, -1/15, 0, 4/15 ],
                                          [ 0, 1/3, 2/3, -2/3, 1/3, 1/3, 0, 2/3 ] ])]>
                         where w := K.1 where K := CyclotomicField(15);
// G21
elif i eq 21 then
  W:=MatrixGroup<2, K |
  [Matrix(K, 2, 2,[ [ 0, -1/3, 0, 0, 0, 2/3, 0, 2/3, 0, 1/3, 0, 1/3, 0, -2/3, 0, -2/3 ],
                    [ 0, -1/3, 0, 0, 0, -4/3, 0, 2/3, 0, 1/3, 0, 1/3, 0, -2/3, 0, 1/3 ],
                    [ 0, -1/15, 0, 0, 0, -4/15, 0, 2/15, 0, 1/15, 0, 1/15, 0, -2/15, 0, 1/15 ],
                    [ 0, 1/3, 0, 0, 0, -2/3, 0, -2/3, 0, -1/3, 0, -1/3, 0, 2/3, 0, 2/3 ] ]),
                         DiagonalMatrix(K, [ 1, w^10 - 1 ])]>
                         where w := K.1 where K := CyclotomicField(60);
// G22
elif i eq 22 then
  W:=MatrixGroup<2, K |
  [Matrix(K, 2, 2,[ [ 0, -4/5, 0, 3/5, 0, -2/5, 0, 1/5 ], [ 0, -2/5, 0, -1/5, 0, -1/5, 0, 3/5 ],
                    [ 0, -2/5, 0, -1/5, 0, -1/5, 0, 3/5 ], [ 0, 4/5, 0, -3/5, 0, 2/5, 0, -1/5 ] ]),
                         Matrix(SparseMatrix(K, 2, 2, [<1, 2, -w>, <2, 1, w^7 - w^5 + w^3 - w>])),
                         Matrix(SparseMatrix(K, 2, 2, [<1, 2, w^7 - w^5 + w^3 - w>, <2, 1, -w>]))]>
                         where w := K.1 where K := CyclotomicField(20);
// G23 = W(H3)
elif i eq 23 then
  W:=MatrixGroup<3, K | [Matrix(K,3,3,[[-1,0,0],[0,1,0],[0,0,1]]),
                         Matrix(K,3,3,[[-1/2*(r^2-3), r/2,0],[r/2, 1/2*(r^2-3),0],[0,0,1]]),
                         Matrix(K,3,3,[[1,0,0],[0,-1/5*(2*r^2 - 5),(2/5)*(2*r^2 - 5)],
                         [0,(2/5)*(2*r^2 - 5),1/5*(2*r^2 - 5)]])]>
                         where r := K.1 where K:=sub<L|w^5*(w^12-w^8)> where w:=L.1 where
                           L := CyclotomicField(20);
                         // note that r^4 = 5*r^2 - 5
// G24
elif i eq 24 then
  W:=MatrixGroup<3, K | [Matrix(K, 3, 3,[ [ 1/2, 0, 0, 0, 0, 0 ], [ 3/14, 3/7, 3/7, 0, 3/7, 0 ],
                                          [ 0, 0, 0, 0, 0, 0 ],
                                          [ -1/2, -1, -1, 0, -1, 0 ], [ -1/2, 0, 0, 0, 0, 0 ],
                                          [ 0, 0, 0, 0, 0, 0 ],
                                          [ 0, 0, 0, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 0 ], [ 1,0, 0, 0, 0, 0 ] ]),
                         Matrix(K, 3, 3,[ [ 1/2, 0, 0, 0, 0, 0 ], [ -3/14, -3/7, -3/7, 0, -3/7, 0 ],
                         [ 0, 0, 0, 0, 0, 0 ],
```

```
                          [ 1/2, 1, 1, 0, 1, 0 ], [ -1/2, 0, 0, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 0 ],
                          [ 0, 0, 0, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 0 ], [ 1,0, 0, 0, 0, 0 ] ]),
                Matrix(K, 3, 3,[ [ 0, 0, 0, 0, 0, 0 ], [ -4/7, -1/7, -1/7, 0, -1/7, 0 ],
                [ -1/4, -1/4, -1/4, 0, -1/4, 0 ],
                          [ -1, 1/3, 1/3, 0, 1/3, 0 ], [ 1/3, 0, 0, 0, 0, 0 ],
                          [ -5/12, -1/4, -1/4, 0, -1/4, 0 ],
                          [ 0, 2/3, 2/3, 0, 2/3,0 ], [ -4/21, 2/7, 2/7, 0, 2/7, 0 ],
                          [ 2/3, 0, 0, 0, 0, 0 ] ])]>
                where w := K.1 where K := CyclotomicField(7);
// G25
elif i eq 25 then
  W:=MatrixGroup<3, K | [DiagonalMatrix(K, [ 1, 1, w ]),
                Matrix(K, 3, 3,[ [ 2/3, 1/3 ], [ -1/3, 1/3 ], [ -1/3, 1/3 ], [ -1/3, 1/3 ],
                [ 2/3, 1/3 ],
                          [ -1/3, 1/3 ], [ -1/3, 1/3 ], [ -1/3, 1/3 ], [ 2/3, 1/3 ] ]),
                DiagonalMatrix(K, [ 1, w, 1 ])]>
                where w := K.1 where K := CyclotomicField(3);
// G26
elif i eq 26 then
  W:=MatrixGroup<3, K | [Matrix(SparseMatrix(K, 3, 3, [<1, 1, 1>, <2, 3, 1>, <3, 2, 1>])),
                DiagonalMatrix(K, [ 1, 1, w ]),
                Matrix(K, 3, 3,[ [ 2/3, 1/3 ], [ -1/3, 1/3 ], [ -1/3, 1/3 ], [ -1/3, 1/3 ],
                [ 2/3, 1/3 ],
                          [ -1/3, 1/3 ], [ -1/3, 1/3 ], [ -1/3, 1/3 ], [ 2/3, 1/3 ] ])]>
                where w := K.1 where K := CyclotomicField(3);
// G27
elif i eq 27 then
  W:=MatrixGroup<3, K | [Matrix(K,3,3,[[ 1/2, 0, 0, 0, 0, 0, 0, 0 ],
                     [ 3/10, 1/5, 1/5, 2/5, -3/5, 1/5, 1/5, -1/5 ],
                     [ 3/10, -3/10, -3/10, -1/10, -1/10, 1/5, 1/5, -7/10 ],
                     [1/2, 1/5, -2/5, 1/5, -3/5, 1/5, 3/5, -4/5 ],
                     [ -1/10, 0, 1/5, -1/5, 0, 0, 0, 1/5 ],
                     [ -1/10, 1/2, 7/10, -7/10, 1/2, 0, 0, 7/10 ],
                     [ 0, -1/5, 2/5, 2/15, -1/15, 2/15, -4/15, 2/15 ],
                     [ 4/15, -1/3, 2/15, -2/15, -1/3, 0, 0, 2/15 ],
                     [ 3/5, 0, -1/5, 1/5, 0, 0, 0, -1/5 ] ]),
                Matrix(K, 3, 3,[ [ 0, 0, 0, 0, 0, 0, 0, 0 ], [ -1/5, -1/5, 0, 0, 3/5, -1/5, 0, 2/5 ],
                          [ 4/5, -1/5, -1/2, 1/2, -2/5, 3/10, 1, -11/10 ],
                          [ -3/5, -1/5, 1/5, -3/5, 3/5, -1/5, -4/5, 3/5 ],
                          [ 2/5, 0, 1/5, -1/5, 0, 0, 0, 1/5],
                          [ 2/5, 0, -3/10, 3/10, 0, 1/2, 0, -3/10 ],
                          [ -1/15, -2/15, 7/15, -1/15, -4/15, 1/5, -8/15, 1/15 ],
                          [ -1/15, 0, -1/5, 1/5, 0, -1/3, 0, -1/5 ],
                          [ 3/5, 0, -1/5, 1/5, 0, 0, 0, -1/5 ] ]),
                DiagonalMatrix(K, [ -1, 1, 1 ])]>
                where w := K.1 where K := CyclotomicField(15);
// G28 = W(F4)
elif i eq 28 then
  W:=MatrixGroup<4, RationalField() | DiagonalMatrix(RationalField(), 4, [1,1,1,-1]),
                          Matrix(RationalField(), 4, 4, [[1,0,0,0],[0,1,0,0],[0,0,0,1],[0,0,1,0]]),
                          Matrix(RationalField(), 4, 4, [[1,0,0,0],[0,0,1,0],[0,1,0,0],[0,0,0,1]]),
                          (1/2)*Matrix(RationalField(), 4, 4, [[1,1,1,-1],[1,1,-1,1],[1,-1,1,1],[-1,1,1,1]])>;
// G29
elif i eq 29 then
  W:=MatrixGroup<4, K | [DiagonalMatrix(K, [ 1, 1, 1, -1 ]),
                Matrix(K,4,4,[[1/2, 0 ], [1/2, 0 ], [0, 1/2 ], [ 0, 1/2 ], [ 1/2, 0 ],
                [ 1/2, 0 ], [ 0, -1/2 ], [ 0, -1/2 ],
                     [0, -1/2 ], [0, 1/2 ], [1/2, 0 ], [ -1/2, 0 ], [ 0, -1/2 ],
                     [ 0, 1/2 ], [ -1/2, 0 ], [ 1/2, 0 ] ]),
                Matrix(SparseMatrix(K, 4, 4, [<1, 2, 1>, <2, 1, 1>, <3, 3, 1>, <4, 4, 1>])),
                Matrix(SparseMatrix(K, 4, 4, [<1, 1, 1>, <2, 3, 1>, <3, 2, 1>, <4, 4, 1>]))]>
                where w := K.1 where K := CyclotomicField(4);
// G30 = W(H4)
elif i eq 30 then
  W:=MatrixGroup<4, K | [Matrix(K,4,4,[[-1,0,0,0],[0,1,0,0],[0,0,1,0],[0,0,0,1]]),
                Matrix(K,4,4,[[-1/2*(r^2-3), r/2,0,0],[r/2, 1/2*(r^2-3),0,0],
                     [0,0,1,0],[0,0,0,1]]),
                Matrix(K,4,4,[[1,0,0,0],[0,-1/5*(2*r^2 - 5),(2/5)*(2*r^2 - 5),0],
                     [0,(2/5)*(2*r^2 - 5),1/5*(2*r^2 - 5),0],[0,0,0,1]]),
                Matrix(K,4,4,[[1,0,0,0],[0,1,0,0],[0,0,-1/2*(r^2 - 2),1/2*(r^3-3*r)],
                     [0,0,1/2*(r^3- 3*r),1/2*(r^2 - 2)]])]>
                where r := K.1 where K:=sub<L|w^5*(w^12-w^8)> where w:=L.1
                  where L := CyclotomicField(20);
                // note that r^4 = 5*r^2 - 5
// G31
elif i eq 31 then
  W:=MatrixGroup<4, K | [DiagonalMatrix(K, [ -1, 1, 1, 1 ]),
                Matrix(SparseMatrix(K, 4, 4, [<1, 2, w>, <2, 1, -w>, <3, 3, 1>, <4, 4, 1>])),
                Matrix(SparseMatrix(K, 4, 4, [<1, 2, 1>, <2, 1, 1>, <3, 3, 1>, <4, 4, 1>])),
                Matrix(K, 4, 4,[ [ 1/2, 0 ], [ -1/2, 0 ], [ -1/2, 0 ], [ -1/2, 0 ],
                [ -1/2, 0 ], [ 1/2, 0 ], [ -1/2, 0 ],
                          [ -1/2, 0 ], [ -1/2, 0 ], [ -1/2, 0 ], [ 1/2, 0 ],
                          [ -1/2, 0 ], [ -1/2, 0 ], [ -1/2, 0 ],
                          [ -1/2, 0 ], [ 1/2, 0] ]),
                Matrix(SparseMatrix(K, 4, 4, [<1, 1, 1>, <2, 3, 1>, <3, 2, 1>, <4, 4, 1>]))]>
                where w := K.1 where K := CyclotomicField(4);
// G32
elif i eq 32 then
```

```
  W:=MatrixGroup<4, K | [DiagonalMatrix(K, [ 1, 1, w, 1 ]),
                        Matrix(K,4,4,[[2/3, 1/3 ], [-1/3, 1/3 ], [ -1/3, 1/3 ], [ 0, 0 ],
                        [ -1/3, 1/3 ], [ 2/3, 1/3 ], [ -1/3, 1/3 ],
                                      [ 0, 0 ], [-1/3, 1/3 ], [-1/3, 1/3 ], [2/3, 1/3 ],
                                      [ 0, 0 ], [ 0, 0 ], [ 0, 0 ], [0, 0 ], [1, 0 ]]),
                        DiagonalMatrix(K, [ 1, w, 1, 1 ]),
                        Matrix(K, 4, 4,[ [ 2/3, 1/3 ], [ 1/3, -1/3 ], [ 0, 0 ], [ 1/3, -1/3 ],
                        [ 1/3, -1/3 ], [ 2/3, 1/3 ], [ 0, 0 ],
                                         [ -1/3, 1/3 ], [ 0, 0 ], [ 0, 0 ], [ 1, 0 ], [ 0, 0 ],
                                         [ 1/3, -1/3 ], [ -1/3, 1/3 ], [ 0, 0 ],
                                         [ 2/3, 1/3 ]])]>
                        where w := K.1 where K := CyclotomicField(3);
// G33
elif i eq 33 then
  W:=MatrixGroup<5, K | [Matrix(SparseMatrix(K, 5, 5, [<1, 2, 1>, <2, 1, 1>, <3, 3, 1>, <4, 4, 1>, <5, 5, 1>])),
                        Matrix(SparseMatrix(K, 5, 5, [<1, 1, 1>, <2, 5, 1>, <3, 3, 1>, <4, 4, 1>, <5, 2, 1>])),
                        Matrix(SparseMatrix(K, 5, 5, [<1, 1, 1>, <2, 2, 1>, <3, 3, 1>, <4, 5, 1>, <5, 4, 1>])),
                        Matrix(SparseMatrix(K, 5, 5, [<1, 1, 1>, <2, 2, 1>, <3, 3, 1>, <4, 5, -w - 1>, <5, 4, w>])),
                        Matrix(K, 5, 5,[ [ 2/3, 0 ], [ -1/3, 0 ], [ 1/3, 0 ], [ -1/3, 0 ],
                        [ -1/3, 0 ], [ -1/3, 0 ], [ 2/3, 0 ],
                                         [ 1/3, 0 ], [ -1/3, 0 ], [ -1/3, 0 ], [ 2/3, 0 ],
                                         [ 2/3, 0 ], [ 1/3, 0 ], [ 2/3, 0 ],
                                         [ 2/3, 0 ], [ -1/3, 0 ], [-1/3, 0 ], [ 1/3, 0 ],
                                         [ 2/3, 0 ], [ -1/3, 0 ], [ -1/3, 0 ],
                                         [ -1/3, 0 ], [ 1/3, 0 ], [ -1/3, 0 ], [ 2/3, 0 ] ])]>
                        where w := K.1 where K := CyclotomicField(3);
// G34
elif i eq 34 then
  W:=MatrixGroup<6, K | [Matrix(SparseMatrix(K, 6, 6, [<1, 2, 1>, <2, 1, 1>, <3, 3, 1>, <4, 4, 1>,
                                                      <5, 5, 1>, <6, 6, 1>])),
                        Matrix(SparseMatrix(K, 6, 6, [<1, 1, 1>, <2, 6, 1>, <3, 3, 1>, <4, 4, 1>,
                        <5, 5, 1>, <6, 2, 1>])),
                        Matrix(SparseMatrix(K, 6, 6, [<1, 1, 1>, <2, 2, 1>, <3, 3, 1>, <4, 4, 1>,
                        <5, 6, 1>, <6, 5, 1>])),
                        Matrix(SparseMatrix(K, 6, 6, [<1, 1, 1>, <2, 2, 1>, <3, 3, 1>, <4, 4, 1>,
                        <5, 6, -w - 1>, <6, 5, w>])),
                        Matrix(K, 6, 6,[ [ 2/3, 0 ], [ -1/3, 0 ], [ 1/3, 1/3 ], [ 0, -1/3 ],
                        [ -1/3, 0 ], [ -1/3, 0 ],
                                         [ -1/3, 0 ], [ 2/3, 0 ], [ 1/3, 1/3 ], [ 0, -1/3 ],
                                         [ -1/3, 0 ], [ -1/3, 0 ],
                                         [ 0, -1/3 ], [ 0, -1/3 ], [ 2/3, 0 ], [ 1/3,1/3 ],
                                         [ 0, -1/3 ], [ 0, -1/3 ],
                                         [ 1/3, 1/3 ], [ 1/3, 1/3 ], [ 0, -1/3 ], [ 2/3, 0 ],
                                         [ 1/3, 1/3 ], [ 1/3, 1/3 ],
                                         [ -1/3, 0 ], [ -1/3, 0 ], [ 1/3, 1/3 ], [ 0, -1/3 ],
                                         [ 2/3, 0 ], [ -1/3, 0 ],
                                         [ -1/3, 0 ], [ -1/3, 0 ], [ 1/3, 1/3 ], [ 0, -1/3 ],
                                         [ -1/3, 0 ], [ 2/3, 0 ] ]),
                        Matrix(SparseMatrix(K, 6, 6, [<1, 1, 1>, <2, 2, 1>, <3, 4, w>, <4, 3, -w - 1>,
                        <5, 5, 1>, <6, 6, 1>]))]>
                        where w := K.1 where K := CyclotomicField(3);
// G35 = W(E6)
elif i eq 35 then
  W:=MatrixGroup<6, RationalField() |
           Matrix(RationalField(), 6, 6, [ -1, 0, 1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0,
                                    0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 6, 6,
                                    [ 1, 0, 0, 0, 0, 0, 0, -1, 0, 1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0,
                                    0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 6, 6,
                                    [ 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 0, -1, 1, 0, 0, 0, 0,
                                    0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 6, 6,
                                    [ 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1,
                                      1, -1, 1, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 6, 6,
                                    [ 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0,
                                      1, 0, 0, 0, 0, 0, 1, -1, 1, 0, 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 6, 6,
                                    [ 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0,
                                      1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 1, -1 ]) >;
// G36 = W(E7)
elif i eq 36 then
  W:=MatrixGroup<7, RationalField() | Matrix(RationalField(), 7, 7,
                                     [ -1, 0, 1, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0,1, 0, 0, 0, 0, 0,
                                        0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0,
                                                                 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 7, 7,
                                    [ 1, 0, 0, 0, 0, 0, 0, 0, -1, 0, 1, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0,
                                         0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0,
                                                                 0, 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 7, 7,
                                    [ 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 1, 0, -1, 1, 0, 0, 0, 0,
                                         0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0,
                                                                 0, 0, 0, 0, 0, 0, 1 ]),
                                    Matrix(RationalField(), 7, 7,
                                    [ 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0,
                                      1, 1, -1, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0,
```

```
                                                                 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 7, 7,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0,
                                        0, 0, 1, 0, 0, 0, 0, 0, 0, 1, -1, 1, 0, 0, 0, 0, 0, 0, 1, 0,
                                           0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 7, 7,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0,
                                           0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1, -1, 1, 0,
                                                                 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 7, 7,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0,
                                            0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0,
                                                                 0, 1, -1 ]) >;
  // G37 = W(E8)
  elif i eq 37 then
    W:=MatrixGroup<8, RationalField() | Matrix(RationalField(), 8, 8,
                                          [ -1, 0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0,
                                             0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0,
                                             0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 8, 8,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 0, -1, 0, 1, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0,
                                           0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0,
                                           1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 8, 8,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 1, 0, -1, 1, 0, 0, 0,
                                           0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1,
                                           0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 8, 8,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0,
                                           0, 0, 1, 1, -1, 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0,
                                           1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 8, 8,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0,
                                          0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, -1, 1, 0, 0, 0, 0, 0, 0, 0, 1,
                                          0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 8, 8,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0,
                                              0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, -1,
                                                 1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 8, 8,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0,
                                               0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1,
                                                   0, 0, 0, 0, 0, 0, 0, 1, -1, 1, 0, 0, 0, 0, 0, 0, 0, 1 ]),
                                         Matrix(RationalField(), 8, 8,
                                         [ 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0,
                                            0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1,
                                       0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 1, -1 ]) >;
  end if;
  return W;
end function;
```

Appendix II: Auxiliary functions

As we will be interested in W-orbits of points in $\mathbf{P}(V)$ and invariants of singularities, we have written a file `auxiliary-functions.m` (given below) which contains several functions:

- `ProjectiveOrbit(grp,pt)` **:** Given a matrix group `grp` in $\mathbf{GL}_n(K)$ and a point `pt` of $\mathbf{P}^{n-1}(K)$, the function returns the orbit of `pt` under `grp`, as a list of elements of $\mathbf{P}^{n-1}(K)$.
- `ProjectiveOrbits(grp,list)` **:** Given a matrix group `grp` in $\mathbf{GL}_n(K)$ and a list of points `list` in $\mathbf{P}^{n-1}(K)$, the function returns the list of orbits under `grp` which meet `list`: the function returns a list of pairs `<pt,m>` where `pt` runs over a set of representatives of these orbits and `m` is the cardinality of the orbit of `pt`.
- `HypersurfaceMilnorNumber(hyp,pt)` **:** Given a hypersurface `hyp` and a point `pt` on `hyp`, this function returns the local Milnor number of `hyp` at `pt`.
- `HypersurfaceTjurinaNumber(hyp,pt)` **:** Given a hypersurface `hyp` and a point `pt` on `hyp`, this function returns the local Tjurina number of `hyp` at `pt`.
- `HypersurfaceSingularity(hyp,pt)` **:** Given a hypersurface `hyp` and a point `pt` on `hyp`, this function returns `true` if `pt` is an isolated singularity and `false` otherwise and, if `true`, it returns two other values: a normal form of the singularity and its type. It is based on the MAGMA function `NormalFormOfHypersurfaceSingularity`.

```
ProjectiveOrbit:=function(grp,pt)
    local i,j,res,v,w,V,PROJ,grpmod,zgrp;
  zgrp:=Centre(grp);
  zgr:=[w : w in zgrp | IsScalar(w)];
  zgrp:=sub<grp | zgrp>;
  grpmod:=Transversal(grp,zgrp);
  V:=VectorSpace(grp);
  PROJ:=AmbientSpace(Scheme(pt));
  v:=V ! Coordinates(pt);
  res:=[PROJ ! Coordinates(V,v*Transpose(w)) : w in grp];
  return [i : i in Set(res)];
end function;

ProjectiveOrbits:=function(grp,points) local i,j,res,test;
  test:=[i : i in points];
  res:=[];
  while # test gt 0 do
    pt:=test[1];
    orb:=ProjectiveOrbit(grp,pt);
    res:=res cat [<pt,# orb>];
    test:=[i : i in test | (i in orb) eq false];
  end while;
  return res;
end function;

HypersurfaceMilnorNumber:=function(hyp,pt)
    local i,n,res,Aff,hyper,p;
  if IsProjective(hyp) then
    hyper,p := AffinePatch(hyp,pt);
    return $$(hyper,p);
  end if;
  if IsAffine(hyp) then
    Aff:=AmbientSpace(hyp);
    res:=Coordinates(pt);
    n:=Dimension(Aff);
    res:=[Aff.i + res[i] : i in [1..n]];
    res:=Evaluate(MinimalBasis(hyp)[1],res);
    return MilnorNumber(res);
  end if;
end function;

HypersurfaceTjurinaNumber:=function(hyp,pt)
    local i,n,res,Aff,hyper,p;
  if IsProjective(hyp) then
    hyper,p := AffinePatch(hyp,pt);
    return $$(hyper,p);
  end if;
  if IsAffine(hyp) then
    Aff:=AmbientSpace(hyp);
    res:=Coordinates(pt);
    n:=Dimension(Aff);
    res:=[Aff.i + res[i] : i in [1..n]];
    res:=Evaluate(MinimalBasis(hyp)[1],res);
    return TjurinaNumber(res);
  end if;
end function;
```

```
HypersurfaceSingularity:=function(hyp,pt)
    local i,n,res,Aff,hyper,p;
  if IsProjective(hyp) then
    hyper,p := AffinePatch(hyp,pt);
    return $$(hyper,p);
  end if;
  if IsAffine(hyp) then
    Aff:=AmbientSpace(hyp);
    res:=Coordinates(pt);
    n:=Dimension(Aff);
    res:=[Aff.i + res[i] : i in [1..n]];
    res:=Evaluate(MinimalBasis(hyp)[1],res);
    return NormalFormOfHypersurfaceSingularity(res);
  end if;
end function;
```

References

[Bon1] C. BONNAFÉ, *A surface of degree* 24 *with* 1440 *singularities of type* D_4, preprint (2018), `arXiv:1804.08388`.

[Bon2] C. BONNAFÉ, *Some singular curves and surfaces arising from invariants of complex reflection groups*, preprint (2018).

[DGPS] W. DECKER, G.-M. GREUEL, G. PFISTER & H. SCHÖNEMANN, *Singular 4-1-1 — A computer algebra system for polynomial computations*, `http://www.singular.uni-kl.de` (2018).

[Magma] W. BOSMA, J. CANNON & C. PLAYOUST, *The Magma algebra system. I. The user language*, J. Symbolic Comput. **24** (1997), 235-265.

[Mic] J. MICHEL, *The development version of the CHEVIE package of GAP3*, J. of Algebra **435** (2015), 308–336.

[Sar] A. SARTI, *Pencils of Symmetric Surfaces in* $\mathbf{P}^3$, J. Algebra **246** (2001), 429-452.

[Sur] `www.imaginary.org/program/surfer`.

[Thi] U. THIEL, *CHAMP: A Cherednik Algebra Magma Package*, LMS J. Comput. Math. **18** (2015), 266-307.

November 3, 2018

CÉDRIC BONNAFÉ, Institut Montpelliérain Alexander Grothendieck (CNRS: UMR 5149), Université Montpellier 2, Case Courrier 051, Place Eugène Bataillon, 34095 MONTPELLIER Cedex, FRANCE
E-mail : `cedric.bonnafe@umontpellier.fr`